

# Bungee Jumping from a Bridge

A bungee cord acts like a spring. If the cord is stretched past its natural unstretched position, it exerts a force that acts to restore the cord to its original position. This force helps to slow the descent of the jumper before he hits the canyon floor. We will jump from the Royal Gorge bridge, 1053 above the canyon floor, using a 415 foot Air Force bungee cord.

Three forces acting on the jumper:

1. gravity,
2. the restoring force,
3. air resistance.

By making reasonable assumptions about the restoring force and the air resistance, the total force is

$$F = -32m + .39(-L - h) - v$$

where  $m$  is the mass of the jumper,  $L$  is the length of the bungee cord, and  $v$  is the velocity.

The level of the bridge is  $h = 0$  and the bottom of the gorge is  $h = -1053$ .

In the following equations, fill in the length of the cord that you wish to use to jump from the bridge (This is  $L$ ). Also, fill in your weight for  $w$ .

```
> L := 415:
> w := 150:
> m := w/32:
```

We want to find how fast the jumper is travelling when the cord engages. Before the cord engages, this is a free fall problem.

$$F = ma - v$$

```
> freefall := diff(s(t),t$2) = -32 - diff(s(t),t)/m;
```

$$\text{freefall} := \frac{\partial^2}{\partial t^2} s(t) = -32 - \frac{16}{75} \left( \frac{\partial}{\partial t} s(t) \right)$$

```
> ffsoln := dsolve({freefall,s(0)=0,D(s)(0)=0},s(t));
```

$$\text{ffsoln} := s(t) = -\frac{5625}{8} e^{(-\frac{16}{75} t)} - 150t + \frac{5625}{8}$$

```
> ff := unapply(op(2,ffsoln),t);
```

$$ff := t \rightarrow -\frac{5625}{8} e^{(-\frac{16}{75}t)} - 150t + \frac{5625}{8}$$

```
> endofrope := fsolve(ff(t)=-L,t);
      endofrope := 6.207229239
```

Depending on your choice of cord, it will take a few seconds to reach the end of your rope. To find the speed at the end of your rope, we plug this value into the derivative.

```
> vf := evalf(subs(t=endofrope, diff(ff(t),t)));
      vf := -110.0980023
```

Since the air resistance also affects the acceleration, we should find the acceleration that the jumper feels at the end of the rope.

```
> af := evalf(subs(t=endofrope,diff(ff(t),t$2)));
      af := -8.512426173
```

Now we want to find the position of the jumper for any time  $t$ . We can use Euler's method to help find the height, velocity, and acceleration of the jumpers. Since over a small interval, the height will be the old height plus the change in time times the velocity over that time period, we see how we can develop the following recurrence relations:

```
> tf := endofrope: t0 := tf: h0 := -L: v0 := vf: a0 := af:
> HTable := [t0,h0]:
> VTable := [t0,v0]:
> ATable := [t0,a0]:
```

We will look at the first minute after the rope engages and let the time increase in 0.1 second steps.

```
> n := 180: Delta := 90/n;
      Δ :=  $\frac{1}{2}$ 
> for i from 1 to n-1 do
> t||i := t||(i-1) + Delta:
> h||i := h||(i-1) + Delta*v||(i-1):
> v||i := v||(i-1) + Delta*a||(i-1):
> a||i := (-32*m - 0.39*(L+h||(i-1)) - v||(i-1))/m:
> HTable := HTable,[t||i, h||i]:
> VTable := VTable, [t||i,v||i]:
> ATable := ATable, [t||i,a||i]:
> od:
> with(plots):
```

Now, plot the free-fall, the fall with the bungee and the bottom of the canyon all on the same plot.


```
> bungee := plot({HTable}, t=tf..tf+90,style=POINT,color=blue):
```

```

> ffplot := plot(ff(t),t=0..tf,color=blue,thickness=2):bottom :=
> plot(-1053,t=0..tf+90):

> display({ffplot,bungee,bottom});

```



Okay, what is the maximum speed?


```

> vbungee := plot({VTable}, t=tf..tf+90,style=POINT,color=black):

> ffvplot := plot(diff(ff(t),t),t=0..tf,color=black,thickness=2):


> display({ffvplot,vbungee});

```



What is the maximum acceleration that the jumper undergoes?

```
> abungee := plot({ATable}, t=tf..tf+90, style=POINT,color=red):
> ffaplot := plot(diff(ff(t),t$2),t=0..tf,color=red,thickness=2):
> display({ffaplot,abungee});
```



Are you surprised?